#######################################################################
# SConscript for vega state_tracker

Import('*')

from sys import executable as python_cmd

env = env.Clone()

env.Append(CPPPATH = [
    '#/src/mapi',
    Dir('.'),
])
env.Append(CPPDEFINES = [
    'KHRONOS_DLL_EXPORTS',
])

api_tmp, = env.CodeGenerate(
        target = 'api_tmp.h',
        script = '#src/mapi/mapi_abi.py',
        source = '#src/mapi/vgapi/vgapi.csv',
        command = python_cmd + ' $SCRIPT --printer vgapi --mode app $SOURCE > $TARGET'
)

st_vega = env.ConvenienceLibrary(
    target = 'st_vega',
    source = env.ParseSourceList('Makefile.sources', 'C_SOURCES')
)

Export('st_vega')
